

Profitipps zur Datums- und Zeitberechnung in Excel

von Ignatz Schels

Wer in Excel die Dauer- und den Endtermin von Arbeitspaketen errechnen oder Meilensteine und Aufgaben im Überblick behalten möchte, kommt um das Thema Datums- und Zeitberechnung nicht herum. Das Prinzip dahinter ist zwar einfach, die Tücke steckt jedoch oft im Detail. Dieser Artikel zeigt, wie Datums- und Zeitberechnung in Excel funktionieren, welche Berechnungsfunktionen wichtig sind und wo die Stolpersteine liegen. Die Beispiele können Sie zusammen mit dem Artikel herunterladen.



Ignatz Schels

Selbstständiger DV-Dozent und Journalist, konzipiert und leitet MS Project-Seminare, Projektberater und Autor

Kontakt: info@schels.de

Mehr Informationen unter: www.projektmagazin.de/autoren/

Das Datumsprinzip

Basis aller Datums- und Zeitberechnungen in Excel ist ein interner Kalender, der mit dem 1. Januar 1900 beginnt und dem 31.12.9999 endet. Ein späteres oder früheres Datum, z.B. den 31.12.1899, erkennt Excel nicht und akzeptiert es deshalb auch nicht. Im Unterschied zu früheren Versionen ist der Excel-Kalender ab Version 97 auch bei den Schaltjahren "sattelfest" und hinsichtlich der Zuordnung von Wochentagen zuverlässig.

Der Excel-Kalender

Gibt der Anwender ein Datum ein, vergleicht Excel es zunächst mit dem internen Kalender. Ist es gültig, trägt Excel die serielle Kalenderzahl ein und weist der Zelle ein Datumsformat zu. Ist es nicht gültig, behandelt Excel die eingegebene Zeichenfolge als Texteintrag und stellt sie entsprechend linksbündig dar. Damit Excel die serielle Zahl anzeigt, die dem Datum entspricht, muss man der Zelle lediglich das Zahlenformat "Standard", "Zahl" oder "0" für eine serielle Ganzzahl zuweisen.

Eingabevarianten

In der Interpretation der Datumseingabe ist Excel sehr flexibel: Alles, was irgendwie nach Datum aussieht, wird akzeptiert. Es genügt bereits die Eingabe zweier Zahlen, die durch einen Punkt getrennt sind, damit Excel sie als Datum interpretiert. Besonders bei Einsteigern, die gewohnt sind Dezimalpunkte zu schreiben, sorgt dies oft für Verwirrung.

Auch internationale Datumsschreibweisen werden richtig interpretiert – die US-amerikanische mit Schrägstrichen ebenso wie die mit Bindestrichen als Trennzeichen zwischen Tag, Monat und Jahr. Nur die Reihenfolge muss der eines deutschen Datums entsprechen.

Diese Technik hat schon viele Excel-Einsteiger zur Verzweiflung getrieben, die versucht haben, einen Bruchwert (12/5, 3/8) oder eine Ordnungszahl mit Bindestrich als Trennzeichen einzugeben und immer nur Datumswerte erhalten haben. Tabelle 1 zeigt einige Beispiele für mögliche Datumsangaben.

Zweistellige Jahreswerte

Ein Besonderheit gilt es zu beachten: Die Eingabe einer zweistelligen Jahreszahl wird je nach Wertigkeit als Zukunfts- oder Vergangenheitswert interpretiert:

Eingabe	Anmerkung	Ausgabe
1.1.1900	Das gültige Datum wird erkannt, Excel trägt die Zahl "1" ein und formatiert die Zelle mit dem Standard-Datumsformat (TT.MM.JJJJ).	"01.01.1900"
12.5.06	Ebenfalls ein gültiges Datum. Die eingetragene serielle Zahl lautet "38849" (38849ter Tag ab dem 1.1.1900).	"12.05.2006"
29.2.06	Kein gültiges Datum, da 2006 kein Schaltjahr ist. Excel stellt das Datum linksbündig dar, als Zeichen, dass es als Texteintrag behandelt wird.	"29.2.06" (kein Datum)
12.5	Excel interpretiert die Eingabe zweier durch einen Punkt getrennter Zahlen als Datum	"12. Mai" (des aktuellen Jahres)
12/5	Die US amerikanische Schreibweise mit Schrägstrich wird als Datum erkannt. Die Reihenfolge muss der eines deutschen Datums entsprechen.	"12. Mai" (des aktuellen Jahres)
1-4-6	Die Trennung durch einen Bindestrich akzeptiert Excel ebenfalls bei einer Datumseingabe. Die Reihenfolge muss der eines deutschen Datums entsprechen.	"01.04.2006"

Tabelle 1: Erkennt Excel ein gültiges Datum, weist es der Zelle automatisch ein Datumsformat zu.

17.12.29 = 17. Dezember 1929 (nicht 2029)

12.07.06 = 12. Juli 2006 (nicht 1906)

Die Grenze zwischen historischem und zukünftigem Datum zieht nicht Excel selbst, sondern eine Einstellung in den "Regions- und Sprachoptionen" der Systemsteuerung. Wählen Sie dort die Registerkarte "Regionale Einstellungen" und klicken Sie auf "Anpassen", um die Kalendereinstellung unter "Datum" zu kontrollieren. Die Registerkarte zeigt übrigens auch, welches Zahlenformat für Standard-Datumswerte verwendet wird (Bild 1).

Datumsformat zuweisen

Damit die serielle Kalenderzahl, mit der Excel rechnet, als Datum angezeigt wird, muss ihr ein entsprechendes Zahlenformat zugewiesen werden. Excel wählt dafür automatisch das Standard-Datumsformat. Unter *Format / Zellen / Register "Zahlen" / Kategorien "Datum"* können Sie den betreffenden Zellen jedoch auch ein anderes Format zuweisen. Alternativ lässt sich die Darstellung mit Hilfe von Platzhalterzeichen in der Kategorie "Benutzerdefiniert" individuell festlegen (Bild 2 und Tabelle 2).




Bild 1: Die Datumseinstellungen werden in der Systemsteuerung festgelegt.

Rechnen mit Datum und Zeit

Wird ein gültiger Zeitwert in Excel eingegeben, z.B. 12:30, konvertiert das Programm den Wert automatisch in eine Dezimalzahl und weist ihr das Format hh:mm zu (Tabelle 3). Eine Stunde entspricht dabei dem Dezimalwert 0,0416667 (= 1 Tag / 24 Stunden). Umgekehrt können Sie den Dezimalwert einer Zeitangabe mit 24 multiplizieren, um die Zeit in dezimaler Schreibweise zu erhalten (z.B.: 12:30 = 0,5208333*24 = 12,5).

Diese Darstellung wird auch als Industriezeit bezeichnet. 12:30 entspricht 12 Industriestunden und 50 Industrieminuten oder 12,5 Industriestunden.

Rechnen mit Datumswerten

Bild 3 zeigt anhand eines Beispiel-Projektplans, wie mit Datumswerten gerechnet wird: Das Enddatum im Plan (Spalte E) ergibt sich durch eine einfache Addition der Werte für "Beginn" (Zelle C6) und "Dauer"

Platzhalter	Ergebnis
T	Tageszahl ohne führende Null
TT	Tageszahl mit führender Null
TTT	abgekürzter Wochentag (Mo, Di, Mi ...)
TTTT	ausgeschriebener Wochentag (Montag, Dienstag ...)
M	Monatszahl ohne führende Null
MM	Monatszahl mit führender Null
MMM	Monat abgekürzt (Jan, Feb, Mär ...)
MMMM	Monat ausgeschrieben (Januar, Februar ...)
J oder JJ	zweistellige Jahreszahl
JJJ oder JJJJ	vierstellige Jahreszahl

Tabelle 2: Platzhalter für Datumsformate.




Bild 2: Das Zahlenformat sorgt dafür, dass aus der seriellen Zahl ein Datum wird Bild

(Zelle D6). Die Dauer muss dabei als Dezimalzahl eingegeben werden. Damit zusätzlich der Text "Tage" erscheint, markieren Sie die Spalte und weisen Sie ihr über *Format / Zellen / Zahlen / Benutzerdefiniert* das Zahlenformat "0" Tage" zu.

Umgekehrt lässt sich die Dauer eines Vorgangs oder einer Projektphase aus der Differenz zwischen Beginn- und Enddatum

berechnen. Damit der erste Tag mit berücksichtigt wird, müssen Sie zum Ergebnis 1 addieren (Bild 4). Wie Sie die tatsächlichen Urlaubstage, d.h. die Differenz abzüglich Wochenend- und Feiertage berechnen, lesen Sie im Abschnitt "Datumsfunktionen".

Arbeits- und Pausenzeiten berechnen

Mit Zeitwerten wird vor allem in Ressourcenberichten gerechnet, wenn es um die Zuteilung von Arbeitszeit geht. Das Beispiel eines Dienstplans in Bild 5 zeigt, wie sich Arbeitszeiten durch einfache Subtraktion berechnen lassen.

Soll zusätzlich auch mit Überstunden, Tarif- bzw. Sollstunden oder Pausenzeiten gerechnet werden, ist meist eine Umrechnung in Industriezeit erforderlich. In unserem Beispiel soll eine Pausenzeit abgezogen werden, und zwar 15 Minuten, wenn die Arbeitszeit weniger als 9 Stunden beträgt und 30 Minuten, wenn sie darüber liegt. Die Pausenzeit wird zunächst über eine WENN-Funktion berechnet. Für die Zelle E6 in Bild 6 lautet die Formel $=\text{WENN}(\text{D6} \times 24 < 9; 15; 30)$. Das Zahlenformat für die Spalte E ist "Standard".

Zur weiteren Berechnung muss das Ergebnis in Zelle E6 in einen Minuten-Zeitwert umgewandelt werden, d.h. durch 24 und anschließend durch 60 geteilt werden, bevor es vom ursprünglichen Ergebnis abgezogen werden kann. Die entsprechende Formel sehen Sie oben in Bild 6. Spalte F erhält wieder das Zeitformat hh:mm.

Vorsicht bei Negativzeiten

Wer das Prinzip der Zeitberechnung kennt, wird auch mit Negativzeiten kein Problem haben. Das Beispiel in Bild 7 zeigt den zuvor erstellten Dienstplan mit zusätzlichen Einträgen für Schichtarbeiter. Da die Endzeit technisch gesehen vor der Anfangszeit liegt, erfordert die Berechnung der Arbeitszeit eine Umweg: Mit einer zusätzlichen WENN-Funktion wird zunächst geprüft, ob der zweite Wert größer ist als der erste. Ist das der Fall, wird die erste Zeit von einem ganzen Tag abgezogen und das Ergebnis zum zweiten Wert addiert.

Stundensumme über 24 Stunden

In unserem Beispiel soll zusätzlich die Summe der Arbeitszeiten aller Mitarbeiter ermittelt werden (Bild 8). Summiert man die Werte auf, scheinen im Ergebnis 48 Stunden zu fehlen, denn Excel zeigt statt 51 nur 3

Eingabe	Anmerkung	Ausgabe
12:30	Ein gültiger Zeitwert, Excel weist nach der Eingabe das Zahlenformat hh:mm zu.	"12:30"
23:9	Ebenfalls ein gültiger Zeitwert, das Zahlenformat fügt die fehlende Null ein.	"23:09"
14:80	Kein gültiger Zeitwert, aber Excel schreibt die Dezimalzahl in die Zelle. Mit der Zuweisung des Zahlenformats hh:mm wird daraus eine gültige Zeit.	"15:20"

Tabelle 3: Beispiele für die Eingabe von Zeitwerten.

E6	fx =C6+D6				
	A	B	C	D	E
1		Projekt:	Neubau Motorenwerk		
2		Projektleiter:	I. Schels		
3					
4					
5	Nr.	Vorgang	Beginn	Dauer	Ende
6	1	Keller ausschachten	03.07.2006	32 Tage	04.08.2006
7	2	Keller betonieren	07.08.2006	38 Tage	
8	3	Fundament betonieren	01.09.2006	20 Tage	
9	4	Halle aufstellen	15.09.2006	30 Tage	
10	5	Dach eindecken	10.10.2006	15 Tage	
11					

Bild 3: Der Endtermin berechnet sich aus der Addition von Anfangstermin und

D6	fx =C6-B6+1			
	A	B	C	D
1	Urlaubsübersicht			
2	Projektteam "Motorenwerk"			
3				
4				
5	Name	Beginn	Ende	Anzahl Tage
6	Heinz Huber	04.07.2006	10.07.2006	7 Tage
7	Friedrich Gröbel	24.07.2006	21.08.2006	29 Tage
8	Manuela Gärtner	01.07.2006	03.07.2006	3 Tage
9	Siegfried Krömer	20.07.2006	20.08.2006	32 Tage

Bild 4: Die Dauer berechnet sich aus der Differenz von End- und Anfangstermin.

D6	fx =C6-B6			
	A	B	C	D
1	Dienstplan			
2				
3				
4				
5	Name	Arbeitsbeginn	Arbeitsende	Arbeitszeit
6	Berger	07:30	16:30	09:00
7	Fröhlich	08:00	17:30	09:30
8	Grundeisen	06:00	12:00	06:00
9	Baumann	09:00	18:00	09:00

Bild 5: Zur Berechnung der Arbeitszeit reicht eine einfache Subtraktion.

Stunden an. Der Grund liegt im Zahlenformat: Excel errechnet als Stundensumme 2 Tage und 3 Stunden, das Zahlenformat hh:mm kann aber nur Stunden anzeigen.

Mit Hilfe eines Sonderformats lösen Sie das Problem: Geben Sie in die Ergebniszelle D13 das benutzerdefinierte Format [h]:mm ein, damit Excel das korrekte Ergebnis "51:00" anzeigt (Bild 8).




Bild 8: Das richtige Zahlenformat für Stundensummen.

Datumsfunktionen

Für aufwändigere Berechnungen von Datums- und Zeitwerten stellt Excel eine Vielzahl an Funktionen zur Verfügung. So gibt es z.B. eine Funktion, die das Tagesdatum aus der Systemzeit berechnet, eine weitere zeigt an, wie viele Tage bis zum Monatsende verbleiben und eine dritte berechnet den Wochentag eines Datums. Die Liste aller verfügbaren Funktionen finden Sie unter *Einfügen / Funktion* im Funktions-Assistenten in der Kategorie "Datum & Zeit" (Bild 9). Tabelle 4 zeigt eine Übersicht der Funktionen.

! Achten Sie darauf, dass die Analyse-Funktionen als Add-In eingeschaltet sind, damit Funktionen wie NETTOARBEITSTAGE, ARBEITSTAG oder MONATSENDE zur Verfügung stehen. Wählen Sie dazu im Menü *Extras* den Unterpunkt *Add-In* und setzen Sie ein Kreuz vor das Add-In "Analyse-Funktionen" (bei internationalen Office-Installationen heißt es *Analysis Toolpack*). Damit stehen die Funktionen im Funktions-Assistenten zur Verfügung. Wird das Add-In wieder deaktiviert oder werden Tabellen mit Analyse-Funktionen in Systemumgebungen ohne dieses Add-In benutzt, melden die Formeln den Fehlerwert "#WERT".




Bild 9: Die Datums- und Zeitfunktionen im Funktions-Assistenten.

Das Tagesdatum

Für das Tagesdatum, das aus der Systemzeit des Computers übernommen wird, stellt Excel gleich zwei Funktionen zur Auswahl. Beide benötigen zwar Funktionsklammern, aber keine Argumente zwischen diesen. Wenn Sie eine dieser Funktionen verwenden, aktualisiert Excel beim Öffnen der Arbeitsmappe und mit jeder Neuberechnung der Tabelle das Datum bzw. Datum und Uhrzeit.

=HEUTE() Das Tagesdatum in der Form TT.MM.JJJJ

=JETZT() Ebenfalls das Tagesdatum, zusätzlich wird aber noch die aktuelle Zeit angezeigt. Das Zahlenformat dafür lautet: TT.MM.JJJJ hh:mm

Die Funktion DATUM()

Diese Funktion wird für die Berechnung von Datumswerten verwendet, wenn die Argumente (Jahr, Monat, Tag) aus anderen Tabellenbereichen bezogen werden:

=DATUM(Jahr;Monat;Tag)

Bild 10 zeigt eine praktische Anwendung dieser Funktion: einen dynamischen Terminkalender, der auch bei einer Änderung der Jahreszahl automatisch die richtigen Datumswerte berechnet. Die Jahreszahl wird aus der Zelle A1 bezogen, der Monat berechnet sich aus dem angezeigten Monat, für die Tageswerte steht die Zeilennummer bereit (abzüglich 2, denn das erste Datum steht schon in Zeile 3). Die Formel wird bis zum Monatsende nach unten kopiert, mit Hilfe der bedingten Formatierung lassen sich die Wochentage farbig hervorheben.

Die Funktion NETTOARBEITSTAGE()


Mit der einfachen Differenz zwischen zwei Datumswerten ist es in Projektplänen meist nicht getan. Häufig werden statt der Kalendertage die Nettotage, d.h. die Wochentage von Montag bis Freitag gezählt und Urlaubszeiten, Feiertage, projektfreie Tage oder Ausfallzeiten berücksichtigt. Die Analyse-Funktion **NETTOARBEITSTAGE()** erledigt diese Aufgabe:

=NETTOARBEITSTAGE(Ausgangsdatum;Enddatum;FreieTage)

In den ersten beiden Argumenten wird das Ausgangs- und Enddatum angegeben. Das dritte Argument ist wahlfrei, hier kann eine Liste von Zellbezügen oder ein Bereich angegeben werden, der z.B. eine Liste mit den Feiertagen des Projektjahres enthält. Das Beispiel zeigt einen Phasenplan, in dem die Nettotage der einzelnen Phasen ermittelt werden.

Funktion	Beschreibung
ARBEITSTAG()	Liefert das Datum vor oder nach einer Anzahl von Arbeitstagen
BRTEILJAHRE()	berechnet den Zeitraum zwischen zwei Datumswerten in Jahren
DATUM()	liefert die serielle Zahl aus einem Datumstext
DATWERT()	wandelt einen Text in ein Datum um
EDATUM()	liefert die Datumszahl eines Tages, der n Monate vom Ausgangsdatum entfernt ist
HEUTE()	Liefert das aktuelle Tagesdatum
JETZT()	Liefert das Tagesdatum und die aktuelle Uhrzeit
JAHR()	wandelt eine serielle Zahl in eine Jahreszahl um
KALENDERWOCHE()	wandelt ein Datum in die Kalenderwoche um
MINUTE()	wandelt eine serielle Zahl in Minuten um
MONAT()	wandelt eine serielle Zahl in einen Monat um
MONATSENDE()	liefert den letzten Tag des Monats, der n Tage vom Ausgangsdatum entfernt ist
NETTOARBEITSTAGE()	liefert die Anzahl Arbeitstage, die zwischen zwei Datumswerten liegen
SEKUNDE()	wandelt eine Zahl in eine Sekunde um
STUNDE()	wandelt eine Zahl in eine Stunde um
TAG()	wandelt eine serielle Zahl in eine Tageszahl um
TAGE360()	berechnet die Anzahl der zwischen zwei Tagen liegenden Tage ausgehend von 360 Tagen pro Jahr
WOCHENTAG()	wandelt eine fortlaufende Zahl in einen Wochentag um
ZEIT()	liefert die serielle Zahl eines Zeitwertes
ZEITWERT()	wandelt einen Text in eine Zeit um

Tabelle 4: Funktionen der Kategorie "Datum & Zeit" und deren Bedeutung.



	A3	B	C	D	E
1	2006				
2	Januar	Termin			
3	01 So				
4	02 Mo				
5	03 Di				
6	04 Mi				
7	05 Do				
8	06 Fr				
9	07 Sa				
10	08 So				
11	09 Mo				

Bild 10: Ein Terminkalender, der mit der Funktion DATUM() erstellt wurde.

Die Funktion ARBEITSTAG()

Eine weitere Datumsfunktion aus dem Paket der Analyse-Funktionen ist ARBEITSTAG(). Sie berechnet das Enddatum, das sich aus der Addition von Kalendertagen zu einem Starttermin ergibt. Im Unterschied zur normalen Addition von Tagen berücksichtigt die Funktion nur Wochentage und schließt wie schon die zuvor beschriebene Funktion auf Wunsch freie Tage aus (projektfreie Tage, Feiertage). Bild 12 zeigt den Phasenplan mit dem berechneten Endtermin "brutto" (Addition der Dauer) sowie dem Endtermin "Arbeitstage" (Funktion ARBEITSTAG()).

Die Funktion WOCHENTAG()

Mit dieser Funktion ermitteln Sie den Wochentag eines Datums als Zahl zwischen 1 und 7. Als Argument wird das Datum übergeben, ein weiteres, nicht benötigtes Argument würde regeln, ab wann gezählt wird. Ohne Angabe dieses Arguments beginnt die Zählung mit dem Sonntag (Wochentagszahl 1). Wendet man die Funktion z.B. auf

ein Datum an, das auf einen Donnerstag fällt, würde man als Ergebnis den Wert 5 erhalten.

Das folgende Beispiel zeigt, wie diese Funktion in der Praxis eingesetzt wird: Der Projektstart ist fixiert, die Konzeptphase soll 23 Tage dauern. Um den Beginn der nachfolgenden Phase zu berechnen, addiert man den Wert für die Dauer der Konzeptphase zum Startdatum (Bild 13). Dabei stellt man fest, dass das Datum auf einen Samstag fällt. Praxisgerechter wäre es, den Termin auf den darauf folgenden Montag zu verschieben. Dazu wird die Addition mit Hilfe der Funktionen WENN() und WOCHENTAG() abgesichert:

$\text{=WENN(WOCHENTAG(B1+B2)=7;B1+B2+2;B1+B2)}$

Mit einer weiteren WENN-Funktion wird auch der Sonntag als mögliches Ergebnis berücksichtigt, die Funktionen werden dazu ineinander geschachtelt:

$\text{=WENN(WOCHENTAG(B1+B2)=7;B1+B2+2;WENN(WOCHENTAG(B1+B2)=1;B1+B2+1;B1+B2))}$

Kalenderwochen berechnen

Zu den häufigen Aufgaben in der Projektarbeit mit Excel gehört die Berechnung der Kalenderwoche. Excel bietet für diese Aufgabe zwar die Funktion KALENDERWOCHE() an, diese rechnet jedoch nicht korrekt, da sie immer davon ausgeht, dass der 1. Januar eines Jahres zur ersten Kalenderwoche gehört. Ist dieser jedoch ein Freitag, Samstag oder Sonntag, beginnt die 1. Kalenderwoche laut DIN 1355 erst in der darauf folgenden Woche. Die Funktion liefert in diesem Fall ein falsches Ergebnis.

Die Kalenderwoche muss also berechnet werden und dazu gibt es eine etwas ausführliche Formel (Bild 14). Vorausgesetzt, das Datum steht in A1, berechnet folgende Formel die zugehörige Kalenderwoche:

$\text{=KÜRZEN((A1-WOCHENTAG(A1;2)-DATUM(JAHR(A1+4-WOCHENTAG(A1;2));1;-10))/7)}$

E5	=NETTOARBEITSTAGE(B5;C5)				
	A	B	C	D	E
1	Phasenplan				
2					
3					
4	Phase	Beginn	Ende	Kalendertage	Netto-Tage
5	Design	12.01.2006	16.01.2006	5 Tage	3 Tage
6	Konzeption	15.02.2006	01.03.2006	15 Tage	11 Tage
7	Planungsphase	03.03.2006	20.03.2006	18 Tage	12 Tage
8	Realisation	21.03.2006	21.05.2006	62 Tage	44 Tage
9	Test	25.05.2006	30.06.2006	37 Tage	27 Tage
10	Implementation	01.07.2006	01.08.2006	32 Tage	22 Tage

Bild 11: Die Funktion NETTOARBEITSTAGE() berechnet die Netto-Differenz zwischen zwei Datumswerten.

E5	=ARBEITSTAG(B5;C5)				
	A	B	C	D	E
1	Phasenplan				
2					
3					
4	Phase	Beginn	Dauer	Endtermin (Brutto)	Endtermin (Arbeitstage)
5	Design	12.01.2006	60 Tage	13.03.2006	06.04.2006
6	Konzeption	15.02.2006	80 Tage	06.05.2006	07.06.2006
7	Planungsphase	03.03.2006	80 Tage	22.05.2006	23.06.2006
8	Realisation	21.03.2006	60 Tage	20.05.2006	13.06.2006
9	Test	25.05.2006	50 Tage	14.07.2006	03.08.2006
10	Implementation	01.07.2006	60 Tage	30.08.2006	22.09.2006

Bild 12: Die Funktion ARBEITSTAG() ermittelt den Endtermin durch Addition der Arbeitstage.

B3	=B1+B2	
	A	B
1	Projektstart:	Donnerstag, 01. Juni. 2006
2	Dauer der Konzeptphase:	23
3	Baubeginn:	Samstag, 24. Juni. 2006

Bild 13: Mit Hilfe der Funktion WOCHENTAG () lassen sich Termine vermeiden, die nicht praxisgerecht sind.

D6		=KÜRZEN((C6-WOCHENTAG(C6;2)-DATUM(JAHR(C6+4-WOCHENTAG(C6;2));1;-10))/7)							
	A	B	C	D	E	F	G	H	I
1		Projekt:	Neubau Motorenwerk						
2		Projektleiter:	I. Schels						
3									
4									
5	Nr.	Vorgang	Beginn	(KW)	Dauer	Ende	(KW)		
6	1	Keller ausschachten	03.07.2006	KW 27	32 Tage	04.08.2006	KW 31		
7	2	Keller betonieren	07.08.2006	KW 32	38 Tage	14.09.2006	KW 37		
8	3	Fundament betonieren	01.09.2006	KW 35	20 Tage	21.09.2006	KW 38		
9	4	Halle aufstellen	15.09.2006	KW 37	30 Tage	15.10.2006	KW 41		
10	5	Dach eindecken	10.10.2006	KW 41	15 Tage	25.10.2006	KW 43		
11									

Bild 14: Die Kalenderwoche korrekt berechnen.

Fazit: Rechnen mit Datum und Zeit – eine Excel-Domäne

Excel bietet umfangreiche Möglichkeiten zur Datums- und Zeitberechnung, so dass damit auch anspruchsvolle Aufgaben gelöst werden können. Mit den hier erläuterten Funktionen lassen sich die häufigsten Aufgabenstellungen bei der Projektdokumentation bewältigen. Die vorgestellten Funktionen sind jedoch nur ein Teil des Angebots, das Excel in der Funktionsliste bereit hält und der geübte Excel-Anwender zaubert mit Funktionen und Formelkonstrukten Analysen und Statistiken aller Art in der Projektdokumentation.

Literatur

- "Excel Formeln und Funktionen", Ignatz Schels, Verlag Markt und Technik, München, 2005